Simple_OTA_VCR

* dsauersanjose@aol.com 10/21/08
* www.idea2ic.com




```
VIN         VIN         0      SIN(    0     10m    1000    )  AC 10m
R0          VIN         OUT    10K
R1          INN         0      100
R2          VOUT        INN    1000k
C1          VOUT        INN    .1u
B_OTA1      OUT         0      I =    -1*v(VIABC)*tanh(( -v(INN) )/.052)
B_BUF1      VOUT        0      V =     v(OUT)
V_Iabc      VIABC       0      PWL     ( 0 1m 10m 0 )

.control
tran        1u          10m    0       1u
set         pensize =   2
plot        v(vin)      v(out)         v(inn)
op
ac          dec         10     1       1000K
plot        db(out)     -db(vin)       title Gain2OUT@IABC1000uA
alter       V_Iabc      dc =           100u
ac          dec         10     1       1000K
plot        db(out)     -db(vin)       title Gain2OUT@IABC100uA
.endc

.end
```

==========END_OF_SPICE==========

To Covert PDF to plain text click below
http://www.fileformat.info/convert/doc/pdf2txt.htm
This code works with winspice.

With OTAs, any type of response current to applied input voltages can be realized. By electronically adjusting how the response current derived, any type of impedance can to a certain extent be made. In this case a Voltage controlled capacitor is being made.



Again the buffer is being simulated to have no offset.
A DC blocking cap would be needed for a LM13700.

```
*
*                         -> C_equiv
*    VIN       R0         :                OUT    @ 1000uA  84Hz/10kOhm  = 0.19uF
*     ______/\  /\  /\_:________________   @  100uA  631Hz/10KOhm = .025uF
*     |        \/  \/    :                |
*     |                 ->      IABC= 1mA->0
*    _|_                          |\                  |  |\
*   /_  \              _|________ | \/\/\__________|_|  \
*  //  \  \            \\\  INP    |  /\/\/              |  /|  VOUT
*  \     \//          ______|/        B_BUF1  |/ |
*   \___/          INN  |    _ B_OTA1              |
*    _|_                |                          |
*    ///      R1        |               R2         |
*         _/\  /\  /\_|_________/\  /\  /\_______|
*        _|   \/  \/      |          \/  \/       |
*         ///              |_____||______________|
*                                ||  C1
```

V_Iabc   VIABC   0   PWL   (0 1m 10m 0)

As IABC goes from 1mA to zero in 10msec, the equivalent capacitor is being reduced in value such that it approaches an open circuit.



At 1mA the capacitor appears to be around .2uF. The values of R1 and R2 must not be open circuits DC wise. The circuit will resemble a capacitor until the C1 effectively shorts the OTA's output to its negative input. At this time, the circuit will resemble a resistor whose impedance value tracks the inverse of IABC.



At .1mA the capacitor appears to be around .025uF. When C1 shorts the OTA's output to its negative input, the effective resistance will be ten times higher.